

Using SAS® Graphics to Explore Behavioral Health Cost Risk

Barbara B. Okerson, HMC, Richmond, VA

ABSTRACT

Behavioral health disorders result in significant economic loss through costs associated with inpatient admissions, lost work days and reduction in at-work productivity. Additionally, over 30% of health plan members with complex health issues have at least one behavioral health issue driving cost risk. This paper looks at graphical representations of cost as part of an effort to develop an overall methodology for prediction of future cost risk from selected behavioral health issues.

SAS provides a comprehensive set of graphics tools and procedures as part of the SAS/GRAPH product. Additionally, the SAS/STAT module provides an extensive library of tools for data analysis. Within SAS/STAT, the new ODS graphics applications add value to this ability by providing customizable statistical graphics that display automatically with these procedures when ODS graphics are invoked. This presentation combines the graphics available in SAS/GRAPH with ODS statistical graphics to explore behavioral health cost risk through examination of medical and pharmacy claims, demographics, and self-reported health readiness assessments (HRAs).

The analytic graphics in this paper were developed with version 9.1.3 of SAS executing on a Windows XP platform. SAS Version 9.1 is required for the ODS graphics extensions. The graphics represented in this paper are not platform-specific and can be adapted by both beginning and advanced SAS users.

INTRODUCTION

Health Management Corporation (HMC) is one of the nation's largest, most experienced providers of integrated care and total health solutions. Since 1983, HMC has offered members and clients comprehensive programs and services intended to empower members to take control of their health and see positive results. Through a population-based approach, HMC's disease management programs provide solutions for prevention, chronic condition support, lifestyle management and complex condition care. HMC's mission is to help improve the health and financial outcomes of its clients and members through innovative health solutions that consider every single member at his/her level of care.

Within HMC, the analytics and research group is responsible for the design, development, validation, and implementation of disease management predictive models. Additionally, the analytics and research group provides research, business intelligence, and program evaluation support to measure the effectiveness of the services provided, analyze potential for future products and add to general knowledge of the health management arena. SAS is the major analytic software used by HMC and, as such, all of the graphics in this paper were produced with SAS software in conjunction with this work.

Data used in this paper are test data sets developed for the purpose of behavioral health data modeling and are not intended to represent the health statistics of an actual population or account.

APPLICATION

Through a review of current research combined with clinical consultation, eight impactable behavioral health conditions were identified and defined: 1) alcohol abuse; 2) anxiety disorder; 3) bipolar; 4) depression; 5) eating disorder; 6) other substance abuse; 7) personality disorder; and 8) schizophrenia. A test population of about 100,000 members was examined for prevalence and cost of these conditions, both for the total behavioral health population and for the population of members targeted as part of a management program identifying members with multiple health care issues. Targeted members fell within the top 3% for future high-health-care cost risk and had an expectation of an inpatient admission within the next year. Over 30% of the members selected for participation in this program had at least one behavioral health issue driving cost risk; for 25% of the selected members with a behavioral health issue, behavioral health was the major cost driver. For those with any identified behavioral health indicator, a mean of 31% of the cost resulted from behavioral health.

Based on cost history, predicted future cost, co-morbidities, and other relevant criteria, those with complex health issues that include one or more of the above eight conditions are to be identified either for: 1) behavioral health referral for all care management, or 2) inclusion in an integrated behavioral/physical health disease management program. In order for this process to occur smoothly, a method for predicting future cost needs to be defined.


Issues with current cost risk methodology arose in that it was developed primarily for identification of cost risk for a number of clinical conditions and did not take into account some of the characteristics of behavioral health issues, under representing the cost risk from behavioral health. This paper is not intended as a conclusion to the development of a cost risk model, but rather a graphical exploration of data as a part of this process. Included graphics look at the following:

- Behavioral health dollars distribution;
- Behavioral health admissions over time;
- Proportion of total medical costs;
- Differences in number and timeframe of encounters by age group;
- Geographical distribution of behavioral health referral percentages; and
- Referral rates versus expected rates.

Selected SAS program code is included with each graphical display. Full SAS code for each graphic is available by request from the author.

GRAPHIC 1 – BEHAVIORAL HEALTH DOLLARS OVERALL DISTRIBUTION THROUGH KERNEL DENSITY ESTIMATION

In their simplest form, kernel density estimators are smoothed histograms. While a histogram restricts observations to only one group or bin, kernel estimators allow observations to be included in more than one group. It can be thought of as a window that slides over the data. The kernel function weights the data passing through the window on its relative distance to the center of the window. Smoothness is determined by the window interval. Because dollars are continuous, this provides a better view of this distribution than a regular histogram that forces categorical assignment. Using a data set of health plan members identified as at risk for future behavioral health based on previous care history, the following graphic illustrates the impact of behavioral health on percent of total care dollars.



GRAPHIC 1 SELECTED SAS CODE

ODS statistical graphics are available with selected statistical procedures, allowing for the creation of graphics without the necessity of saving data points to an output data set. These graphics use templates and style elements to create presentation-ready graphics. To modify ODS graphics, modifications need to be made to the template creating the graphic using Proc TEMPLATE. The template can be located by using the ODS TRACE statement, which returns to the SAS log a listing of output objects created. The graph above is created with the Stat.KDE.Graphics.HistDensity object. Once a template name is located, the template code can be located by typing *odst* (or *odstemplates*) on the command line. From the template window, the specific object code was found in the subdirectories under *sashelp.tmplmst*. The Proc TEMPLATE code as modified for the above graphic is shown below.

```
Proc TEMPLATE;
define statgraph Stat.KDE.Graphics.HistDensity;
dynamic _HISTOGRAM _DENSITY _DEPLABEL;
layout Gridded;
  layout overlay / padbottom=5;
  layout gridded / columns=3 valign=top;
  entrytitle _DEPLABEL /fontsize=20 padbottom=5 padtop=5;
  endlayout;
endlayout;
layout Overlay/ XAXISOPTS=(LABEL='Percent') ;
if (_HISTOGRAM=1)
  HistogramParm x=BINX y=BINCOUNT / fillcolor=
  GraphDataDefault:foreground;
endif;
if (_DENSITY=1)
```

```

series x=DENSITYX y=DENSITYCOUNT / linecolor=
StatGraphFitLine:contrastcolor;
endif;
EndLayout;
EndLayout;
end;
run;

```

The Proc KDE code to create the graphic is provided below.

```

ods noproctitle;
ods html path='r:\bokerson\sas global forum\test\' body='kde.htm';
ods graphics on;


Proc KDE data=samplebh;
  univar PCT_Annual_Dollars_BH_Dollars
    / method=sjp plots=histdensity bwm=.8;
run;
ods graphics off;
ods html close;

```

GRAPHIC 2 – TIME SERIES GRAPHIC OF BEHAVIIORAL HEALTH ADMISSIONS

The SAS TIMESERIES procedure was designed to perform trend and seasonal analysis on transactional data with respect to time. The analysis is similar to SAS Proc MEANS, but is applied for each season or time period of concern. Several ODS graphics are available with Proc TIMESERIES including a series plot as illustrated below. The template for the series plot is stored in the Statgraph.TimeSeries.SeriesPlot object.

In this example, behavioral health admissions are plotted for detection of seasonality for a single client. Data is bi-monthly. The graph shows an increase in admissions over time with an upswing each year either at the beginning of the year or just after. The display suggests a need for further investigative analysis to definitively answer the question of seasonality.



GRAPHIC 2 SELECTED SAS CODE

The series plot graphic above uses many of the ODS graphic defaults for this procedure. The only modification to the template was to the entrytitle line which was modified to place a custom title on the graphic:

```
EntryTitle "Behavioral Health Admissions" / padtop=10 padbottom=10 fontsize=20;
```

The code for the timeseries procedure is below. Note the ods noproctitle statement that precedes the ods html call. This is necessary to suppress an overall output title and needs to precede the html call statement.

```

ods noproctitle;
ods html path='r:\bokerson\sas global forum\test\'
  body='timeplot.htm' style=analysis;
ods graphics on;

```

```


Proc TIMESERIES data=admit
  plot=(series);
  id Date interval=month2;
  var Admissions;
run;
ods graphics off;
ods html close;

```

GRAPHIC 3 – PROPORTION OF INPATIENT DOLLARS RELATED TO BEHAVIORAL HEALTH FOR ADMISSIONS WITH ANY BEHAVIORAL HEALTH DIAGNOSIS

When developing a cost risk model, it is important to not only know the behavioral health costs, but how much impact these costs have on total costs, especially in inpatient and emergency department settings. Behavioral health conditions add to inpatient costs regardless of whether the admission is for a behavioral health condition or whether the behavioral health condition is comorbid. This graphic illustrates for patients with the targeted behavioral health conditions, how those conditions drive overall cost.

The graphic below is a bar-bar overlay chart. SAS does not have a procedure that produces these charts, but they can be produced as below by combining the GPLOT procedure with the SAS %bar macro. This type of graphic seems more appropriate than a stacked bar when, although one bar is part of the other, each total is referenced separately. While the graphic shows a decline in behavioral health cost proportion, further analysis would be needed to understand what is driving this change.



GRAPHIC 3 SELECTED SAS CODE

The selected code below shows the creation of annotate data sets for each bar. The legend is also created with an annotate statement – the GPLOT procedure uses three annotate data sets.

```

/*Overall cost bar*/
data anno_bh1;
set b;
by year bh all;
retain xsys ysys '2' when 'a';
%bar(year-.25,0, year+.2,all,CXAAD6FE,0,solid);
run;

/*BH cost bar*/
data anno_bh2;
set b;
by year bh all;
retain xsys ysys '2' when 'a';
%bar(year-.15,0, year+.3,bh,CX77ADFD,0,solid);
run;

/*GPLOT*/
title "Proportion of Inpatient Total Cost that is Behavioral Health Cost"
j=c "(Behavioral Health Patients Only)";

```

```


Proc Gplot data=b anno=anno_legend;
plot all*year/ anno=anno_bh1 overlay noframe nolegend haxis=axis1 vaxis=axis2
    vref=20 to 60 by 20 autovref;
plot2 bh*year / anno=anno_bh2 vaxis=axis3;
run; quit;

```

GRAPHIC 4 - AREA GRAPH OF BEHAVIORAL HEALTH COST BY QUARTER AND AGE GROUP

As a further examination of seasonality, a drill down into additional variables can provide information beyond that of the series plot above. Proc GAREABAR generates bar graphs in which the width of the bars is proportional to a characteristic of the data element that is represented by the bars. The GAREABAR procedure produces a chart based on the values of category variables. Therefore, GAREABAR treats values of numeric category variables as discrete and does not calculate midpoints. Subgroups are supported as in the example below. Data value order follows that of the data set. Area bar charts as produced by GAREABAR are supported only with the *activex* or *actximg* device drivers. The graph below was created using the *activex* driver.

The graphic illustrates that, while the under-18 age group is a smaller percentage of total behavioral health encounters, the distribution by quarter is different for this group with most encounters occurring during the school year.



GRAPHIC 4 SELECTED SAS CODE

Because the *activex* driver is intended for web display, transferring the graphic to Microsoft Word results in a loss of resolution. To overcome this, GOPTIONS was used to oversize the graphic so that a small graphic imported into Word would retain presentation resolution. Colors were also set with GOPTIONS. Proc TEMPLATE was used to set the background color to white. While values are treated as discrete by default, use of the discrete option on the hbar statement was used to eliminate minor tick marks.

```

/*Set GOPTIONS*/
filename odsout 'r:\bokerson\sas global forum\test\gareabar.htm';
ods html file=odsout;
goptions reset=all dev=activex border ftitle='Arial Italic' cback=white
hsize=8.42 vsize=6 ftext='Arial' htext=1.8
    colors=(cxe0ffff cx87cefa cx1e90ff cx0000cd ) transparency ;
title j=c h=2 "One Year Behavioral Health Encounters by Quarter and Age Group";


/*GAREABAR Code*/
Proc GAREABAR data=house_bh;
hbar age_group*n /sumvar=sum subgroup=quarter rstat=pct discrete;
format age_group afmt.;
label sum="Percent of Total Amount Paid" n="Total Encounters" quarter="QTR";
run; quit;

ods html close;

```

GRAPHIC 5 – BEHAVIORAL HEALTH REFERRAL PATTERN MAP

Referrals to behavioral health programs are becoming standard in care management programs because behavioral health issues make it difficult for many to engage in their own care, ultimately adding cost to comorbid conditions. Identifying referral patterns that exist for members who are referred to and are engaged by behavioral health programs adds information that is vital for predicting program participation, cost and outcomes. The map below uses a national data set to identify geographical differences in referrals. The display (county level) shows that successful referrals are more likely in New England and the U.S. Northwest than in other areas. The map was created with the SAS/GRAPH GMAP procedure.



GRAPHIC 5 SELECTED SAS CODE


Because the data set contained zip codes with no county information, the sashelp.zipcode data set (provided by and regularly updated by SAS) was used to map zip codes to counties. Map colors were set in the GOPTIONS statement with the patterns selected providing a progression from white (no cases) to dark purple (top quintile). Selected SAS code for the GMAP procedure is listed below.

```
goptions colors=(white BWH VPAB VLIPB LIB MOPB) ftext='Arial' ctext=black;
legend cborder=gray label=("Quintile   ")
  value=("None" "1" "2" "3" "4" "5" );
title;
title3 h=2 'Behavioral Health Referral Patterns';
Proc GMAP map=maps.uscounty data=bhdatamap;
  id state county;
  choro pctbgrp / legend=legend outline=gray;
run;
quit;
```

GRAPHIC 6 BEHAVIORAL HEALTH DASHBOARD

Another graphical display that can be useful when projecting cost risk for a behavioral health program is a graphic of actual members' participation. The example below uses a dashboard to test the assumptions for engagement by the program against those engaged in the past by other programs. This same type of graphic can be used to track numbers versus assumptions of the new program over time. In the world of business, dashboards are used to organize and present information from multiple sources in a way that the user can understand, especially for tracking goals and outcomes. This is just as relevant for the health care industry. While this graphic is static, the graphic can easily be customized to include drill-downs to data sets and additional information.

This dashboard was created using SAS/GRAPH Proc GCHART and SAS Annotate. Each bullet line is a separate chart – they are joined into a single graphic with Proc GREPLAY. The graphic shows that, although the expected number has been met for referral target, actual enrollments and engagements are not as high as expected.



GRAPHIC 6 SELECTED SAS CODE

The code below was used to create the annotate bars, thick line and expected value line. The code was set up using macro variables so that changes that affect all bars could be made in a single location, making it easier to reuse the code for additional applications.

```

data myanno;
  set temp_data;
  length function $8 color $12 style $20 text $20;
  hsys='3';
  xsys='2'; ysys='1'; style='solid'; when='b';

/*Annotate bars*/
  x=0; y=0; function='move';
  output;
  x=range1; y=100; function='bar'; color="&c1";
  output;

  x=range1; y=0; function='move';
  output;
  x=range2; y=100; function='bar'; color="&c2";
  output;

  x=range2; y=0; function='move';
  output;
  x=range3; y=100; function='bar'; color="&c3";
  output;

/* Annotate a thick line representing the expected value. */
  x=target; y=15; function='move';
  output;
  x=target; y=85; function='draw'; line=1; size=1.7; color="&barc";
  output;

/* Annotate the midpoint value label */
  xsys='1'; ysys='1';
  x=-2; y=85; function='label'; when='a'; position='4'; style="&ftitle";
  size=12; text=trim(left(label1));
  output;
  x=-2; y=30; function='label'; when='a'; position='4'; style="&ftext";
  size=10; text=trim(left(label2));
  output;
run;

```

CONCLUSION

Visual depictions of data relationships assist analysts and other users in better understanding of data through the quick identification of true associations, anomalies, and areas for further study. In the examples in this paper, SAS software, especially SAS/GRAPH and SAS/STAT with the addition of ODS statistical graphics, provided the flexibility and ease of use allowing for visual identification of data relationships; this information provided a foundation for further analyses to identify additional interactions and critical factors that enhance the behavior health risk modeling process.

REFERENCES

HMC website. <http://www.choosehmc.com>.

Rodriguez, R. "An Introduction to ODS for Statistical Graphics in SAS 9.1", *SUGI Proceedings*, 2003.

SAS Institute Inc. Data Visualization: SAS Dashboard Examples.
<http://support.sas.com/rnd/datavisualization/dashboards/>. Accessed December 2007.

SAS Institute Inc. ODS Graphics Overview: ODS Graphics Style Elements and Attributes
<http://support.sas.com/rnd/base/topics/statgraph/proctemplate/a002791342.htm>.
Accessed August 2007.

SAS Institute Inc., *SAS/GRAPH Software: Reference, Version 8, Volumes 1 and 2.*, 1999.

Yeh, S. Customizing ODS Statistical Graphics, *SUGI Proceedings*, 2005.

ACKNOWLEDGEMENTS

I would like to acknowledge the members of the HMC analytic and research group for their suggestions and assistance in the development of this paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. For more information contact:

Barbara B. Okerson, Ph.D., CPHQ
Senior Health Information Consultant
National Account Outcomes
Health Management Corporation (HMC)
8831 Park Central Drive, Suite 100
Richmond, VA 23227
Office: 804-662-5287
Fax: 804-662-5364
Email: bokerson@choosehmc.com



SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.